# Logarithmische Spirale

logarith-spirale.pdf

Prof. Dr. Dörte Haftendorn, Juni 09 Update 19.06.09
**www.mathematik-verstehen.de** **http://haftendorn.uni-lueneburg.de**
**#####################################################################**

```
sp:=plot::Polar([ln(t),t],t=0.001..50,
Mesh=500):plot(sp)
```

Achtung: oft wird die Spirale mit dem Term e^t als logarithmische Spirale bezeichnet. Ich benenne die Spiralen aber nach ihren Termtyp.




Schnitte mit der x-Achse: x=0, x= $\ln(\pi \cdot k)$

```
ln(k*PI)
```

$\ln(\pi \cdot k)$

Der Ursprung wird erreicht für t=1

```
float(tan(1)); float(arctan(%)/PI*180),"Grad"
```

1.557407725

57.29577951, "Grad"

Dies ist der Steigungswinkel der Schlaufe im Ursprung.
Zugehörige kartesische Darstellung

```
plotfunc2d(ln(t), ln(2*k*PI)$
k=1..10,ln(2*9*PI),t=0..20*PI, LegendVisible=FALSE)
```








Hier sieht man, dass sie nach außen zu immer enger wird.

```
float(ln(k*PI/4+2*PI)-ln(k*PI/4)) $ k=1..6
```

2.197224577, 1.609437912, 1.299282984, 1.098612289, 0.955511445, 0.8472978604

Darstellung der Abstände

```
plotfunc2d(ln(t+2*PI)-ln(t),t=0..300, Axes=Origin,
ViewingBoxYRange=0..3)
```




Die Breite der Spiralenringe geht gegen Null. Die Breite der gesamten logarithmischen Spirale ist aber nicht beschränkt.

###################################################

Betrachtung der Fläche

```
iterm:=x->1/2*int(ln(t)^2, t=1..x): iterm(x)
```

$$\frac{x \cdot \left(\ln(x)^2 - 2 \cdot \ln(x) + 2\right)}{2} - 1$$

2

```
1/2*int(ln(t)^2, t=1..PI/2);float(%)
```

$$\frac{\pi \cdot \left(\ln\left(\frac{\pi}{2}\right)^2 - 2 \cdot \ln\left(\frac{\pi}{2}\right) + 2\right)}{4} - 1$$

$$\frac{\pi \cdot \left(\ln\left(\frac{\pi}{2}\right)^2 - 2 \cdot \ln\left(\frac{\pi}{2}\right) + 2\right)}{4} - 1$$

$$0.021615716$$

Dies ist die winzige Fläche im 1. Quadranten
#########################################
Berechnungen für die Schlaufe:
Schnitt mit sich

```
solve(-ln(t)=ln(t+PI),t);
lo:=solve(0=ln((t+PI)*t),t);float(%)
```

$$\mathrm{solve}(-\ln(\pi + t) - \ln(t) = 0, t)$$

$$\left\{-\frac{\pi}{2} - \frac{\sqrt{\pi^2+4}}{2}, \frac{\sqrt{\pi^2+4}}{2} - \frac{\pi}{2}\right\}$$

$$\{-3.432892216, 0.2912995623\}$$

Der zweite Wert erzeugt einen negativen Radius und ist der Parameterwert der Spitze.

```
float(arctan(lo[2])/PI*180);float(%+180);
```

$$16.2408183$$

$$196.2408183$$

```
float(1/2*int(ln(t)^2,t=lo[2]..lo[2]+PI))
```

$$0.9377890457$$

Dies ist der Flächeninhalt der kleinen Schlaufe

```
plotfunc2d(1/2*ln(lo[2])^2,1/2*ln(t)^2,t=lo[2]..lo[2]+PI
,
LegendVisible=FALSE)
```








Hier sieht man nochmals, dass der Winkel richtig berechnet ist,
bei dem die Schlaufe ihre Spritze hat.
Steigungen an dieser Spitze:

```
r:=t->ln(t)
```

$$t \rightarrow \ln(t)$$

```
steig:=t->(r'(t)*sin(t)+r(t)*cos(t))/(r'(t)*cos(t)-r(t)*sin(t))
```

$$t \rightarrow \frac{r'(t) \cdot \sin(t) + r(t) \cdot \cos(t)}{r'(t) \cdot \cos(t) - r(t) \cdot \sin(t)}$$

```
plotfunc2d(steig(t),t=0.01..6)
```






Die einzige Wendestelle ist als Extremstelle gut zu sehen.

Nenner und Zähler extra:

Nenner und Zähler extra:

```
denom(steig(t)), numer(steig(t))
```

$$t \cdot (\cos(t) - t \cdot \ln(t) \cdot \sin(t)),\ t \cdot (\sin(t) + t \cdot \cos(t) \cdot \ln(t))$$

Polarwinkel, bei denen waagerechte Tangenten vorliegen.

```
numeric::solve(numer(steig(t))=0,t=1);
numeric::solve(numer(steig(t))=0,t=2);
numeric::solve(numer(steig(t))=0,t=5);
```

$$\{0.3522153993\}$$

$$\{2.127615825\}$$

$$\{4.84255834\}$$

Polarwinkel, bei denen senkrechte Tangenten vorliegen.

```
numeric::solve(denom(steig(t))=0,t=1);
numeric::solve(denom(steig(t))=0,t=3);
numeric::solve(denom(steig(t))=0,t=6);
```

$$\{1.272850698\}$$

$$\{3.379916142\}$$

$$\{6.367811514\}$$

Spitzenwinkel der Schlaufe:

```
st1:=float(steig(lo[2]));arctan(st1);w1:=
float(%/PI*180);
st2:=float(steig(lo[2]+PI));arctan(st2);w2:=
float(%/PI*180);
w1-w2
```

$$-0.05367861497$$

$$-0.05362714751$$

$$-3.07260922$$

$$-16.82260973$$

$$-1.511422386$$

$$-86.59812377$$

$$83.52551455$$



Der Winkel in der Spitze der Schlaufe ist 83,52°.

```
sps:=plot::Polar([ln(t),t],t=lo[2]..lo[2]+PI,
Mesh=500):plot(sps)
```

```
Mesh=500):plot(sps)
```




Kartesische Kordinaten der Spitze der Schlaufe:

```
xs:=float(r(lo[2])*cos(lo[2]));
ys:=float(r(lo[2])*sin(lo[2]));
```

$$-1.181441637$$

$$-0.3542300216$$